

MISER-II

Andy Stadler

There's treasure, adventure, and more than a little danger in this sequel to the MISER adventure in the first *Commodore 64 Fun and Games* book. We don't want to give away the story, but you'll explore the miser's island (if you can figure out how to get there). There are puzzles to solve, dead ends to avoid, and you may have to learn how to use scuba diving equipment.

Once the program starts, give one- or two-word commands. For example, you might type **NORTH** (or just **N**) to go north. The text of

```
YOU ARE IN THE MISER'S MARINA, STANDING
ON A DOCK.
OBVIOUS EXITS: S E
E
YOU ARE ON THE MISER'S BOAT.
OBVIOUS EXITS: W U D
*
```

MISER-II will be purple (purple prose?) and your responses will be shown in green. Useful commands include **LOOK**, to look around the room you are in; **INVEN**, to get an inventory of your treasures; **SCORE**, to learn your current score and **EXAM**, to look at a specific object.

You might want to make a map of your wanderings if you ever hope to solve the game. Yes, it can be solved! However, like all text adventure programs, it takes lots of patience, and more than a little cleverness to reach the final goal.

```
1 PG$="MISER..II":AU$="BY.ANDY.STADLER":BG$="RETURN" :rem 8100
3 :
4 REM COPYRIGHT (C) 1984 THE CODE WORKS
5 REM BOX 6905, SANTA BARBARA, CA 93160
6 :
10 REM AS OF 25JUL84
90 GOTO 62000 :rem 51784
100 GOTO 44000 :rem 41377
500 IF MB=1 THEN PRINT "{down}YOUR.MATCH.JUST.WENT.OUT." :rem 52137
505 IF MB>0 THEN MB=MB-1 :rem 48789
510 IF AC OR (WS=0) THEN 535 :rem 34245
515 AL=AL-1 :rem 17742
520 IF AL<>0 THEN 530 :rem 61567
525 PRINT "{down}YOU.DRAW.YOUR.LAST.BREATH.FROM.THE.SCUBAGEAR.{down}":
    GOTO 36000 :rem 10090
530 IF AL<5 THEN PRINT "{down}YOU.ARE.DANGEROUSLY.LOW.ON.AIR." :rem 31604
535 IF AC THEN AC=AC-1 :rem 46986
540 IF AC=1 THEN PRINT "{down}AN.ANGEL.QUIETLY.FLOATS.BY.":AC=5 :rem 35505
545 IF DC THEN DC=DC-1 :rem 11557
550 IF DC=1 THEN GOSUB 610 :rem 7596
555 IF ES=0 THEN 605 :rem 9619
560 ES=ES-1:AL=AL+5:IF ES THEN 605 :rem 30440
565 PRINT "{14°space down red}B.O.O.M.!" :rem 12414
570 IF CP<>OL%(6) THEN 580 :rem 47702
575 PRINT "{down}THE.SCUBA.TANK.EXPLODES.IN.YOUR.FACE.{down}":
    GOTO 36000 :rem 39881
580 PRINT "{down}
    A.TREMENDOUS.EXPLOSION.ROCKS.THE.CAVE...YOU.ARE.SHAKEN.BUT.UNHURT."
    :rem 48106
585 Y=OL%(6):FOR X=1 TO OC:IF OL%(X)=Y THEN OL%(X)=0 :rem 63885
590 NEXT X:EX=1:OL%(22)=Y:IF Y<>39 THEN 605 :rem 5065
```

```

595 MV$(40,6)=39:MV$(39,5)=40 :rem 46604
600 RM$(39)="IN^A^ROCKY^ROOM^WITH^LIGHT^POURING^THROUGH^A^BIG^HOLE^IN^THE
    ^ROOF" :rem 20872
605 RETURN :rem 35286
610 PRINT "{down}SINCE^YOU'RE^DEAD,^WHAT^SAY^WE^JUST^PACKIT^IN^NOW,^OK?":
    RETURN :rem 7955
700 GOSUB 500:PRINT "{grn}":SC=0:SF=0:TV=0:GOSUB 60000 :rem 30706
710 PRINT "{pur}":IF LEN(IN$)=0 THEN PRINT "{grn 2°up}";:GOSUB 60000:
    GOTO 710 :rem 35572
720 IF LEFT$(IN$,1)="^" THEN IN$=RIGHT$(IN$,LEN(IN$)-1):
    GOTO 720 :rem 63606
730 IF RIGHT$(IN$,1)="^" THEN IN$=LEFT$(IN$,LEN(IN$)-1):
    GOTO 730 :rem 58202
735 SP=LEN(IN$)+1:IF SP=1 THEN 700 :rem 41265
740 SC=SC+1:IF MID$(IN$,SC,1)="^" THEN SF=SF+1:SP=SC :rem 1409
750 IF SC<LEN(IN$) THEN 740 :rem 8873
760 IF SF>-1 AND SF<2 THEN 780 :rem 31289
770 PRINT "PLEASE^TYPE^A^ONE^OR^TWO^WORD^COMMAND.":GOTO 700 :rem 31535
780 CV$=LEFT$(LEFT$(IN$,SP-1),4):CV=0:CN$="":CN=0 :rem 64992
790 IF SF<>0 THEN CN$=MID$(IN$,SP+1,4) :rem 17563
800 FOR X=1 TO VC:IF CV$=VB$(X) THEN CV=X :rem 362
810 NEXT X:IF CV=0 THEN 31000 :rem 52417
820 IF SF=0 THEN 900 :rem 59496
830 FOR X=1 TO NC:IF CN$=NN$(X) THEN CN=X :rem 5740
840 NEXT X:IF CN=0 AND CV<>17 THEN 32000 :rem 8511
900 ON CV GOTO 1000,1000,2000,2000,3000,3000,4000,4000,5000,5000,6000,
    6000,7000 :rem 51118
910 ON CV-13 GOTO 7000,8000,8000,9000,10000,10000,11000,12000,13000,14000,
    15000 :rem 11000
920 ON CV-24 GOTO 15000,16000,17000,14000,19000,19000,18000,20000,21000,
    22000 :rem 57510
930 ON CV-34 GOTO 23000,21000 :rem 13766
990 PRINT "UNIMPLEMENTED.":END :rem 63720
1000 IF MV$(CP,1)=0 THEN 37000 :rem 21403
1010 CP=MV$(CP,1):TV=1:GOTO 7000 :rem 63808
2000 IF MV$(CP,2)=0 THEN 37000 :rem 11817
2010 IF EX=0 OR CP<>18 THEN 2020 :rem 7212
2015 PRINT "THE^PATH^IS^BLOCKED^BY^A^RECENT^CAVE-IN.":GOTO 700 :rem 48431
2020 CP=MV$(CP,2):TV=1:GOTO 7000 :rem 22678
3000 IF MV$(CP,3)=0 THEN 37000 :rem 21069
3010 CP=MV$(CP,3):GOTO 7000 :rem 49127
4000 IF MV$(CP,4)=0 THEN 37000 :rem 54605
4010 CP=MV$(CP,4):GOTO 7000 :rem 51969
5000 IF MV$(CP,5)=0 THEN 37000 :rem 19639
5010 CP=MV$(CP,5):GOTO 7000 :rem 41591
6000 IF MV$(CP,6)=0 THEN 37000 :rem 11653
6010 CP=MV$(CP,6):GOTO 7000 :rem 14127
7000 IF (CV>12) AND (CV<15) AND CN THEN PRINT "TRY^EXAM.":
    GOTO 700 :rem 19721
7005 OU$="YOU^ARE^"+RM$(CP)+".":GOSUB 30000 :rem 50940
7010 FOR X=1 TO OC:IF OL$(X)<>CP THEN 7030 :rem 64674
7020 OU$="THERE^IS^A^"+OB$(X)+"^HERE.":PRINT:GOSUB 30000 :rem 32339
7030 NEXT X :rem 33789
7100 PRINT "{down}OBVIOUS^EXITS:^":FOR X=1 TO 6 :rem 15829
7110 IF MV$(CP,X)>0 THEN PRINT MID$(DR$,X,1);"^": :rem 18178

```

```

7120 NEXT X:PRINT:IF (TV=0) OR (CP>55) OR (CP<52) OR (OL%(23)=0) THEN
700 :rem 65461
7130 PRINT:OU$="A_STRANGE_VOICE_INTONES,^'" :rem 14482
7140 IF CP=52 THEN OU$=OU$+"HE_WHO_MAKES_ME,_WANTS_ME_NOT.'" :rem 36106
7150 IF CP=53 THEN OU$=OU$+"HE_WHO_HAS_ME,_KNOWS_ME_NOT.'" :rem 18471
7160 IF CP=54 THEN OU$=OU$+"HE_WHO_KNOWS_ME,_WANTS_ME_NOT.'" :rem 47840
7170 IF CP=55 THEN OU$=OU$+"PUT_ME_HERE.'" :rem 54427
7180 GOSUB 30000:GOTO 700 :rem 27887
8000 PRINT "YOU_ARE_CARRYING_THE_FOLLOWING:{down}":FOR X=1 TO OC :rem 61172
8010 IF OL%(X)<>-1 THEN 8200 :rem 24626
8015 PRINT OB$(X) :rem 63658
8020 IF X=5 AND OL%(5)=-1 THEN PRINT "^^^THERE_ARE";ML;"MATCHES_LEFT."
:rem 54900
8030 IF X=6 AND WS THEN PRINT "^^^WHICH_YOU_ARE_WEARING." :rem 25343
8200 NEXT X:IF MB THEN PRINT "BURNING_MATCH" :rem 43954
8210 GOTO 700 :rem 42893
9000 IF SF=0 THEN 34000 :rem 53369
9010 IF ((CN<5) OR (CN>7)) THEN 9100 :rem 3016
9020 PRINT "A_HOLLOW_VOICE_SAYS,^'WRONG_ADVENTURE'." :GOTO 700 :rem 55687
9100 PRINT "OKAY,^'";RIGHT$(IN$,LEN(IN$)-SP);".". :FOR X=1 TO 1000:
NEXT X :rem 16796
9110 PRINT "NOTHING_HAPPENS." :GOTO 700 :rem 16976
10000 IF CN=0 THEN 34000 :rem 36508
10010 IF FNW(CN)=-1 THEN PRINT "YOU_ALREADY_HAVE_IT." :GOTO 700 :rem 15828
10020 IF FNW(CN)<>CP THEN PRINT "I_DON'T_SEE_IT_HERE." :GOTO 700 :rem 21257
10030 IF FNF(CN) THEN PRINT "THAT_ITEM_STAYS_PUT." :GOTO 700 :rem 34543
10040 IF (CN=9) AND ES THEN PRINT "THE_FIRE_HAS_MADE_IT_MUCH_TOO_HOT."
GOTO 700 :rem 254
10050 OL%(PN%(CN))=-1:PRINT "OK" :rem 63012
10060 IF FNT(PN%(CN)) THEN PRINT "{down}YOU_GOT_A_TREASURE!" :rem 26622
10070 GOTO 700 :rem 34416
11000 IF CN=0 THEN 34000 :rem 49964
11010 IF FNW(CN)<>-1 THEN PRINT "YOU_DON'T_HAVE_IT." :GOTO 700 :rem 34228
11012 IF (CN=9) AND WS THEN PRINT "YOU_ARE_WEARING_IT." :GOTO 700 :rem 58165
11015 PRINT "OK" :rem 15008
11020 IF (CN=4) AND MB AND (OL%(10)=CP) THEN 11200 :rem 7359
11025 IF MB THEN 700 :rem 24869
11030 IF CN<>9 OR FB=0 OR OL%(15)<>CP THEN 11040 :rem 32891
11035 PRINT "{down}IT_LANDS_ON_THE_FLAMES." :ES=4 :rem 54654
11040 IF (CP=55) AND (PN%(CN)=23) THEN 11300 :rem 2811
11080 OL%(PN%(CN))=CP :rem 61094
11085 IF (CP=43) AND (CV=31) THEN PRINT "{down}IT_FLUTTERS_AWAY..." :
OL%(PN%(CN))=0 :rem 29021
11090 IF CP<>49 THEN 700 :rem 57995
11100 GOSUB 20500:IF S=100 THEN 21030 :rem 22020
11110 GOTO 700 :rem 57336
11200 PRINT "{down}THE_MATCH_IGNITES_THE_BRUSH.{down}":FB=1:MB=0:
PN%(14)=-15:OL%(15)=CP :rem 37222
11210 OL%(10)=0:GOTO 7000 :rem 62497
11300 OU$="{down}BEFORE_YOUR_VERY_EYES,_THE_MONEY_SHIMMERS_AND_CHANGES..."
:rem 14363
11305 GOSUB 30000 :rem 34248
11310 OL%(23)=0:OL%(8)=55:PN%(27)=8:PN%(28)=8:GOTO 700 :rem 25295
12000 GOSUB 35000 :rem 51308
12040 IF CN=9 THEN PRINT "IT_HAS_A_LABEL_AND_A_GAUGE." :GOTO 700 :rem 7169

```



```

12060 IF CN=13 THEN PRINT "IT_READS";AL*250;"PSI.":GOTO 700 :rem 53954
12070 IF CN=16 THEN PRINT "IT_HAS_A_RED_BUTTON_AND_A_BLUE_BUTTON.":
      GOTO 700 :rem 55490
12090 IF CN=3 THEN 12300 :rem 30439
12100 IF CN=20 THEN 13080 :rem 55791
12110 IF (CN=21) OR (CN=15) OR (CN=4) THEN 12400 :rem 41812
12120 IF PN%(CN)=23 THEN PRINT "IT_APPEARS_TO_BE_COUNTERFEIT.":
      GOTO 700 :rem 51245
12130 IF PN%(CN)=8 THEN PRINT "IT'S_REAL.":GOTO 700 :rem 29717
12190 PRINT "I_SEE_NOTHING_SPECIAL_ABOUT_IT.":GOTO 700 :rem 36109
12300 PRINT "IT'S_AN_AD_FOR_AN_AMUSEMENT_PARK.^^IT" :rem 37688
12310 OU$="SHOWS_A_MAN_THROWING_RINGS_AT_A_BUNCH_OF_BOTTLES.":
      GOSUB 30000 :rem 14966
12320 GOTO 700 :rem 9583
12400 PRINT "THERE'S_SOMETHING_WRITTEN_ON_IT.":GOTO 700 :rem 11445
13000 GOSUB 35000 :rem 32599
13040 IF CN=3 THEN PRINT "IT'S_JUST_A_PICTURE_(TRY_EXAM).":
      GOTO 700 :rem 52047
13050 IF CN=4 THEN 13200 :rem 58353
13060 IF CN=15 THEN 13300 :rem 30607
13070 IF CN=13 THEN 12060 :rem 41498
13080 IF CN=20 THEN PRINT "IT_SAYS,_'WEAR/UNWEAR'.":GOTO 700 :rem 902
13090 IF CN<>21 THEN 13190 :rem 31328
13100 PRINT "IT_SAYS:_'RETURN_TREASURES_HERE_FOR{5°space}POINTS.'":
      GOTO 700 :rem 59529
13190 PRINT "THERE'S_NOTHING_WRITTEN_ON_IT.":GOTO 700 :rem 61897
13200 OU$="IT_SAYS,_'THE_BEST_IN_C64_SOFTWARE_--_THE_CODE_WORKS,_SANTA"
      :rem 4289
13210 OU$=OU$+"_BARBARA,_CA'":GOSUB 30000:GOTO 700 :rem 51987
13300 OU$="THE_LABEL_SAYS,_'COMPRESSED_AIR_^^KEEP_FROM_FIRE_AND_HEAT.'"
      :rem 60577
13310 GOSUB 30000:GOTO 700 :rem 37336
14000 GOSUB 35000 :rem 48783
14010 IF (PN%(CN)=5) AND ML THEN PRINT "OK":ML=ML-1:MB=2:GOTO 700 :rem 6076
14020 IF (PN%(CN)=5) AND (ML=0) THEN PRINT "NO_MATCHES_LEFT.":
      GOTO 700 :rem 62338
14030 IF PN%(CN)=12 THEN PRINT "THE_MATCHES_ARE_TOO_WET_TO_BURN.":
      GOTO 700 :rem 28540
14040 PRINT "YOU_RUB_IT_TOGETHER_UNTIL_YOUR_ARMS_ARE" :rem 58654
14050 OU$="TIRED,_BUT_I_GUESS_YOU_WEREN'T_A_BOY_SCOUT.^^IT_DOESN'T_LIGHT."
      :rem 32390
14060 GOSUB 30000:GOTO 700 :rem 42534
15000 IF CN<>0 THEN 31000 :rem 61874
15010 IF (CP=1) OR (CP=2) OR (CP=5) THEN 15200 :rem 60497
15020 IF CP=42 THEN CP=18:GOTO 15300 :rem 39381
15030 IF CP=18 THEN CP=42:GOTO 15300 :rem 14208
15040 IF CP=32 THEN CP=37:GOTO 15300 :rem 36818
15050 IF CP=37 THEN CP=32:GOTO 15300 :rem 65107
15060 IF (CP<36) AND (CP>32) THEN 15600 :rem 23555
15070 OU$="YOU_EXECUTE_A_PERFECT_SWAN_DIVE_AND_" :rem 16335
15075 OU$=OU$+"HIT_THE_FLOOR_WITH_A_LOUD_SMACK" :rem 43591
15080 GOSUB 30000:PRINT:GOTO 36000 :rem 48557
15200 GOSUB 15400:PRINT "YOU_SWIM_AROUND,_BUT_FIND_NOTHING_OF{4°space}
      INTEREST." :rem 37197
15210 GOTO 15500 :rem 18807

```

```
15300 GOSUB 15400:PRINT "UNDER^THE^SURFACE,^YOU^FIND^A^TUNNEL." :rem 49004
15310 PRINT "ADVENTURER^YOU^ARE,^YOU^FOLLOW^IT.{down}":GOTO 15500 :rem 43198
15400 IF OL%(9)=-1 THEN PRINT "YOU^SINK^LIKE^A^ROCK.{down}":
GOTO 36000 :rem 57262
15405 IF AL THEN IF WS THEN RETURN :rem 53569
15410 OU$="YOU^ARE^NOT^A^FISH.^WITH^NO^AIR^SUPPLY,^YOU^QUICKLY
^DROWN.{down}" :rem 7883
15420 GOSUB 30000:GOTO 36000 :rem 20908
15500 IF OL%(5)=-1 THEN OL%(5)=0:OL%(12)=-1:PN%(4)=12 :rem 18095
15510 IF OL%(10)=-1 THEN OL%(10)=0:OL%(11)=-1:PN%(14)=11 :rem 5502
15520 GOTO 7000 :rem 30229
15600 PRINT "THE^RUSHING^WATER^SWEEPS^YOU^AWAY...{4°space}
YOU^ARE^PUMMELED^BY^THE^"; :rem 20238
15610 PRINT "ROCKS...":GOTO 36000 :rem 44379
16000 GOSUB 35000 :rem 37678
16010 IF CN<>9 THEN PRINT "YOU^WOULDN'T^LOOK^TOO^GOOD^IN^IT.":
GOTO 700 :rem 65180
16020 IF WS THEN PRINT "YOU^ARE^ALREADY^WEARING^IT.":GOTO 700 :rem 53531
16030 WS=1:PRINT "OK":GOTO 700 :rem 64528
17000 GOSUB 35000 :rem 33785
17010 IF (CN<>9) OR (WS=0) THEN PRINT "YOU^AREN'T^WEARING^IT.":
GOTO 700 :rem 39076
17020 PRINT "OK":WS=0:GOTO 700 :rem 13098
18000 IF (CN<>10) OR (CP<>48) THEN 11000 :rem 61872
18010 PRINT
"YOU^RING^A^SPIKE!^^A^LITTLE^TROLL^WALKS^OUT,^CONGRATULATES^YOU,^";
:rem 10948
18020 PRINT "AND^HANDS^YOU^A^PRIZE.":OL%(17)=-1:OL%(7)=0:GOTO 700 :rem 40198
19000 GOSUB 35000 :rem 61760
19010 IF CN=17 THEN 19100 :rem 60334
19015 IF CN=26 THEN PRINT "I^NEED^A^COLOR.":GOTO 700 :rem 41982
19020 IF CN<>18 THEN PRINT "YOU^PUSH^QUITE^HARD,^BUT^TO^NO^AVAIL.":
GOTO 700 :rem 48033
19030 IF MV%(2,4)=1 THEN PRINT "NOTHING^HAPPENS.":GOTO 700 :rem 6163
19040 OU$="THE^ENGINES^ROAR^TO^LIFE,^AND^THE^BOAT^MAKES^A^3^HOUR
^JOURNEY.{down}" :rem 22139
19050 GOSUB 30000:MV%(2,4)=1:GOTO 7000 :rem 3928
19100 IF MV%(2,4)=5 THEN PRINT "NOTHING^HAPPENS.":GOTO 700 :rem 53131
19110 OU$="THE^ENGINES^ROAR^TO^LIFE,^AND^THE^BOAT^MAKES^A^3^HOUR
^JOURNEY.{down}" :rem 43430
19130 GOSUB 30000:MV%(2,4)=5:GOTO 7000 :rem 58465
20000 IF CN THEN 31000 :rem 17074
20010 GOSUB 20500:OU$="WERE^YOU^TO^QUIT^NOW,^YOU^WOULD^HAVE^A^SCORE^OF"
:rem 32509
20015 IF S<1 THEN OU$=OU$+"^0":GOTO 20025 :rem 32307
20020 OU$=OU$+STR$(S) :rem 51230
20025 OU$=OU$+"^POINTS.":GOSUB 30000:
PRINT "{down}THIS^PUTS^YOU^IN^THE^CLASS^OF:" :rem 63255
20030 PRINT "{down}<";RT$;"^ADVENTURER>":
PRINT "{down}DO^YOU^INDEED^WISH^TO^QUIT^NOW?{grn down}" :rem 36518
20040 GOSUB 60000:IF LEFT$(IN$,1)<>"Y" THEN PRINT "{pur down}OK":
GOTO 700 :rem 59477
20050 PRINT "{pur down}THANKS^FOR^A^FUN^GAME!{down}":GOTO 21070 :rem 3460
20500 S=0:FOR X=1 TO OC:IF NOT FNT(X) THEN 20530 :rem 64008
20510 IF OL%(X)=49 THEN S=S+20 :rem 1152
```

```

20520 IF OL%(X)=-1 THEN S=S+5 :rem 5946
20530 NEXT X:IF AC THEN S=S-20 :rem 29064
20540 FOR X=1 TO CC:IF S<RP%(X) THEN RT%=RT%(X):RT=X:RETURN :rem 6582
20550 NEXT X:RETURN :rem 35650
21000 IF CN THEN 31000 :rem 57522
21010 PRINT "DO YOU REALLY WANT TO QUIT?{grn down}":GOSUB 60000 :rem 34406
21020 IF LEFT$(IN$,1)<>"Y" THEN PRINT "{down pur}OK":GOTO 700 :rem 38087
21030 GOSUB 20500:PRINT "{pur down}YOU ACCUMULATED A SCORE OF",S,
      "POINTS." :rem 61760
21040 PRINT "{down}THIS PUTS YOU IN A CLASS OF:{down}":
      PRINT "<","RT$","ADVENTURER>{down}" :rem 51624
21050 IF RT<>CC THEN PRINT "BETTER LUCK NEXT TIME..." :rem 51728
21060 IF RT=CC THEN PRINT "CONGRATULATIONS!" :rem 16185
21070 FOR X=1 TO 6000:NEXT:GOTO 60600 :rem 30663
22000 GOSUB 35000 :rem 9105
22010 PRINT "IT WON'T A BUDGE.":GOTO 700 :rem 65316
23000 IF CN THEN 31000 :rem 34184
23010 IF CP<>43 THEN PRINT "CAREFUL! YOU MIGHT HIT YOUR HEAD!":
      GOTO 700 :rem 4921
23020 IF OL%(1)=-1 THEN GOSUB 23100 :rem 18721
23030 PRINT "YOU AREN'T A BIRD! IT'S A LONG WAY....":PRINT:
      GOTO 36000 :rem 4901
23100 OU$="I GUESS YOU NEVER LEARNED TO RE-PACK PARACHUTES.{down}"
      :rem 24796
23110 GOSUB 30000:RETURN :rem 63385
30000 OE=40:IF LEN(OU$)<40 THEN PRINT OU$:RETURN :rem 22033
30010 IF MID$(OU$,OE,1)<>"^" THEN OE=OE-1:GOTO 30010 :rem 53320
30020 PRINT LEFT$(OU$,OE-1):PRINT RIGHT$(OU$,LEN(OU$)-OE):RETURN :rem 64814
31000 EM=3-EM:PRINT EM$(EM):GOTO 700 :rem 35096
32000 OU$="I DON'T KNOW WHAT A^"+RIGHT$(IN$,LEN(IN$)-SP)+"^IS.":
      GOSUB 30000 :rem 6982
32010 GOTO 700 :rem 9590
33000 PRINT "I DON'T SEE IT HERE.":GOTO 700 :rem 53917
34000 PRINT IN$;"^WHAT?":GOTO 700 :rem 56079
35000 IF CN=0 THEN 34000 :rem 33866
35010 IF FNW(CN)=-1 THEN RETURN :rem 46548
35020 IF FNW(CN)<>CP THEN PRINT "I DON'T SEE IT HERE.":GOTO 700 :rem 1323
35030 IF FNR(CN) THEN PRINT "YOU DON'T HAVE IT.":GOTO 700 :rem 50264
35040 RETURN :rem 30112
36000 CP=46:AC=6:DC=9:GOTO 7000 :rem 21091
37000 PRINT "THERE IS NO WAY TO GO IN THAT DIRECTION.":GOTO 700 :rem 52110
40000 DATA 55,36,28,23,6 :rem 12759
40010 DATA 0,49,2,0,0,0,"IN THE MISER'S MARINA, STANDING ON A DOCK"
      :rem 4959
40020 DATA 0,0,0,1,3,4,"ON THE MISER'S BOAT" :rem 6475
40030 DATA 0,0,0,0,0,2,"AT THE HELM OF THE MISER'S BOAT" :rem 21324
40040 DATA 0,0,0,0,2,0,"BELOW DECKS ON THE MISER'S BOAT" :rem 13950
40050 DATA 0,0,2,0,6,0 :rem 52702
40055 DATA "IN THE MISER'S PRIVATE ISLAND MARINA, STANDING ON A DOCK"
      :rem 46255
40060 DATA 0,0,7,8,0,5,"ON A LONG PATH OVERLOOKING THE MISER'S MARINA"
      :rem 36304
40070 DATA 0,43,42,6,0,0,"STANDING AT A FORK IN THE PATH" :rem 17293
40080 DATA 0,12,6,9,0,0 :rem 56319

```


40085 DATA "STANDING_ON_THE_FRONT_LAWN_OF_THE_MISER'S_SUMMER_HOUSE"
:rem 24560

40090 DATA 8,10,0,0,0,0,"ON_THE_WEST_SIDE_OF_THE_HOUSE" :rem 41118

40100 DATA 0,41,11,9,0,0,"IN_THE_MISER'S_BACK_YARD" :rem 63240

40110 DATA 8,10,0,0,0,17,"ON_THE_EAST_SIDE_OF_THE_HOUSE" :rem 13995

40120 DATA 8,0,0,0,0,0,"ON_THE_MISER'S_FRONT_PORCH" :rem 11139

40130 DATA 12,14,15,50,0,0,"IN_THE_LIVING_ROOM" :rem 13802

40140 DATA 13,0,0,0,0,17,"IN_THE_TINY_KITCHEN" :rem 13820

40150 DATA 0,16,0,13,0,0,"IN_THE_BEDROOM" :rem 52575

40160 DATA 15,0,0,0,0,0,"SQUEEZED_INTO_THE_CLOSET" :rem 36859

40170 DATA 0,0,11,0,14,0,"IN_THE_DARK_CELLAR" :rem 6504

40180 DATA 0,19,0,0,0,0 :rem 45521

40185 DATA "IN_A_GIANT_BLUIISH-WALLED_CAVERN,_AT_THE_EDGE_OF_A_MURKY_POOL"
:rem 33196

40190 DATA 18,0,48,20,0,0,"AT_THE_EAST_END_OF_A_STRANGE_CORRIDOR" :rem 9483

40200 DATA 0,52,19,30,0,21 :rem 13794

40205 DATA "IN_A_STRANGE_CORRIDOR_WHICH_STRETCHES_OUT_OF_SIGHT" :rem 35140

40210 DATA 22,0,0,0,20,0,"ON_A_STEEP_PATH" :rem 31242

40220 DATA 23,21,0,0,0,0,"IN_A_CHEESY_SMELLING_ROOM" :rem 22527

40230 DATA 0,25,0,22,0,26,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 14975

40240 DATA 25,0,0,23,0,0,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 38026

40250 DATA 24,0,26,0,27,0,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 61074

40260 DATA 27,25,0,24,0,0,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 32354

40270 DATA 26,28,28,28,28,28,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 26790

40280 DATA 27,0,29,27,27,27,"IN_A_STRANGE,_CONFUSING_ROOM" :rem 46899

40290 DATA 28,0,0,0,0,0,"IN_A_SPARKLING_CHAMBER" :rem 50014

40300 DATA 44,0,20,0,0,31,"AT_THE_WEST_END_OF_THE_STRANGE_CORRIDOR"
:rem 35871

40310 DATA 32,0,0,0,30,0 :rem 39335

40315 DATA "AT_A_PLACE_WHERE_THE_SLOPE_FLATTENS_OUT_AND_HEADS_NORTH"
:rem 42351

40320 DATA 0,31,0,0,0,0,"IN_A_LOW_RED_ROOM_BORDERING_A_DEEP_POOL" :rem 10021

40330 DATA 0,36,0,0,35,34 :rem 50004

40335 DATA
"IN_A_STEEP-WALLED_CAVERN._A_SWIRLING_RIVER_RUSHES_BY_AT_YOUR_FEET"
:rem 31550

40340 DATA 0,0,0,0,33,0 :rem 5052

40345 DATA "DOWNSTREAM_IN_THE_RIVER_CANYON._THE_WATER_APPEARS_DEEPER_HERE"
:rem 32414

40350 DATA 0,0,0,0,0,33,"UPSTREAM._THE_CANYON_WALLS_ONLY_LOOM_EVEN_HIGHER"
:rem 25251

40360 DATA 33,37,0,38,0,0,"IN_A_TRIANGULAR_SHAPED_ROOM" :rem 49939

40370 DATA 36,51,0,45,38,0,
"IN_A_REDDISH_WALLED_CAVERN,_EDGING_A_MURKY_LAKE" :rem 32982

40380 DATA 36,0,0,0,39,37,
"IN_A_NONDESCRIPT_ROOM._COOL_AIR_BLOWS_FROM_ABOVE" :rem 49145

40390 DATA 0,0,0,0,0,38 :rem 31899

40395 DATA "IN_A_ROCKY_ROOM_WITH_LIGHT_PEEKING_THROUGH_A_HOLE_IN_THE_ROOF"
:rem 14087

40400 DATA 0,0,0,41,0,0,"STANDING_ON_A_ROCKY_DOME" :rem 26285

40410 DATA 10,0,40,0,0,0,"TRUDGING_ALONG_A_CURVED,_UPHILL_PATH" :rem 37348

40420 DATA 0,0,0,7,0,0,"AT_THE_WEST_END_OF_A_DEEP,_INVITING_LAKE" :rem 21055

40430 DATA 7,0,0,0,0,0,"STANDING_AT_THE_EDGE_OF_A_GIANT_CLIFF" :rem 55404

40440 DATA 0,30,0,0,0,0,"STANDING_IN_A_TIGHT_ALCOVE" :rem 18718

```

40450 DATA 0,0,37,0,0,0,"IN_A_SQUARE_ROOM" :rem 23540
40460 DATA 47,47,47,47,47,47,"RESTING_ON_A_CLOUDBANK_ABOVE_MIAMI" :rem 64227
40470 DATA 46,46,46,46,46,46,"AT_THE_PEARLY_GATES" :rem 55425
40480 DATA 0,0,0,19,0,0 :rem 24827
40485 DATA "IN_THE_HALL_OF_ICICLES.^STRANGE_SPIRES_COVER_THE_FLOOR"
:rem 26351
40490 DATA 1,0,0,0,0,0,"IN_A_STORAGE_CLOSET" :rem 13223
40500 DATA 0,0,13,0,0,0,"IN_THE_HALL_CLOSET" :rem 57738
40510 DATA 37,0,0,0,0,0,"IN_A_ROUND_ROOM" :rem 54956
40520 DATA 20,53,0,0,0,0,"IN_A_LONG_NARROW_TUBE" :rem 45405
40530 DATA 52,54,0,0,0,0,"IN_A_LONG_NARROW_TUBE" :rem 7620
40540 DATA 53,55,0,0,0,0,"IN_A_LONG_NARROW_TUBE" :rem 53562
40550 DATA 54,0,0,0,0,0,"AT_THE_END_OF_THE_TUBE" :rem 46005
41000 DATA N,NORT,S,SOUT,E,EAST,W,WEST,U,UP,D,DOWN,L,LOOK,I,INVE,SAY,GET,
TAKE :rem 36669
41010 DATA DROP,EXAM,READ,LIGH,DIVE,SWIM,WEAR,UNWE,IGNI,PUSH,PRES,THRO,SCOR
:rem 14072
41020 DATA QUIT,OPEN,JUMP,Q :rem 41400
42000 DATA PARA,1,BUCK,2,BROC,4,MATC,5,XYZZ,-13,PLUG,-13,RITN,-13,DOOR,-3
:rem 24023
42010 DATA SCUB,6,RING,7,KEY,8,GOLD,9,GAUG,-6,BRUS,10,LABE,-6,AUTO,-14,RED,
-14 :rem 57047
42020 DATA BLUE,-14,GIRA,17,GRAF,-16,SIGN,-19,RUBI,20,CUBE,20,MINK,21,COAT,
21 :rem 50653
42030 DATA BUTT,-14,CASH,23,MONE,23 :rem 41206
43000 DATA PARACHUTE,-1,BUCKET,-1,DOOR,12,"TRAVEL_BROCHURE" :rem 52835
43005 DATA 21,"PRINTED_MATCHBOOK" :rem 22957
43010 DATA 35,"SCUBA_EQUIPMENT",16 :rem 57303
43015 DATA "SIX_INCH_PLASTIC_RING",45,"*$10,000*",0 :rem 32538
43020 DATA "*HEAVY_NUGGET_OF_GOLD*",34,"DRY_PILE_OF_BRUSH" :rem 3950
43025 DATA 51,"SOGGY_PILE_OF_BRUSH",0 :rem 11701
43030 DATA "SOGGY_MATCHBOOK",0,<$>,0,AUTOPILOT,3,"RAGING_BRUSH_FIRE",0
:rem 32207
43040 DATA "GRAFFITI_COVERING_THE_WALL",9,"*STUFFED_GIRAFFE*",0 :rem 33713
43050 DATA "CELLAR_STAIRWAY_LEADING_DOWN",11,"SIGN_ON_THE_WALL",49 :rem 4424
43060 DATA "*PLATINUM_RUBIK'S_CUBE(TM)*",29,"*MINK_COAT*",50 :rem 22175
43070 DATA "SCORCHED_AREA_ON_THE_WALL",0,"$10,000",4 :rem 10250
43500 DATA NOVICE,5,AMATEUR,30,EXPERIENCED,50,PRO,70,MASTER,90,GRANDMASTER,
999 :rem 11661
44000 READ RC,VC,NC,OC,CC :rem 51252
44010 DIM RM$(RC),MV$(RC,6),VB$(VC),NN$(NC),PN$(NC),OB$(OC),OL$(OC),RT$(CC)
:rem 34063
44015 DIM RP$(CC):PRINT "{wht}SETTING_UP...{pur}" :rem 41972
44020 FOR X=1 TO RC:FOR Y=1 TO 6:READ MV$(X,Y):NEXT Y:READ RM$(X):
NEXT X :rem 25016
44030 FOR X=1 TO VC:READ VB$(X):NEXT X :rem 64301
44040 FOR X=1 TO NC:READ NN$(X),PN$(X):NEXT X :rem 26178
44050 FOR X=1 TO OC:READ OB$(X),OL$(X):NEXT X :rem 18553
44055 FOR X=1 TO CC:READ RT$(X),RP$(X):NEXT X :rem 62565
44060 CP=1:DR$="NSEWUD":EM$(1)="WHAT?":EM$(2)="I_DON'T_UNDERSTAND_THAT.":
EM=1 :rem 39592
44070 PRINT "{clr}":ML=4:AL=11 :rem 246
44080 DEF FNW(X)=OL$(ABS(PN$(X))) :rem 53855
44090 DEF FNR(X)=(PN$(X)>0) :rem 31297
44100 DEF FNF(X)=(PN$(X)<0) :rem 63041

```



```
44110 DEF FNT(X)=(LEFT$(OB$(X),1)="*") :rem 47206
44999 GOTO 7000 :rem 21854
```

389 lines, proof number = 36299

Reminder: Now be sure to enter the standard framework lines 60000 to 62200. See page XV.

Important Variables in MISER-II

RM\$() Room descriptions	SC Spaces counter
MV%() Move maps	SF Spaces found
VB\$() Verbs	CV\$ Command verb
NN\$() Nouns	CV Command verb number
PN\$() Noun to object pointers	CN\$ Command noun
OB\$() Objects	CN Command noun number
OL%() Object locations	ML Matches left
EM\$() Error messages	MB Match burning
RT\$() Ratings	FB Fire burning
RP%() Ratings point values	DD Dead
RC Room count	DC Dead message counter
VC Verb count	AC Angel counter
NC Noun count	WS Wearing scuba
OC Object count	AL Air left
RC Ratings count	ES Explosion sequence
RT\$ Rating given	EX Explosion happened
CP Current position	S Score
EM Current error position	TV Trigger voice

How MISER-II Works

500-610 Background processes	22000-22010 Open
700-840 Parse the input and look up words	23000-23110 Jump
900-990 Jump tables to command routines	30000-30020 Output Routine
1000-1010 North	31000 Error "What?"
2000-2020 South	32000-32010 Error "What is a..."
3000-3010 East	33000 Error "Not here..."
4000-4010 West	34000 Error "(Verb) what?"
5000-5010 Up	35000-35040 Check availability of object
6000-6010 Down	36000 Dead
7000-7180 Look	37000 Can't go that way
8000-8210 Inventory	40000 How many rooms, verbs, nouns, objects, score, classes
9000-9110 Say	40010-40550 Room data: destinations for N,S,E,W,U,D, and room descriptions
10000-10070 Get	41000-41020 Verbs
11000-11110 Drop	42000-42030 Noun and pointer to its object. Positive number is a "real" object (can be picked up). Negative number is a fixed object (exists but can't be taken).
12000-12400 Exam	43000-43070 Objects and the room they start in
13000-13310 Read	43500 Score classes, title, and minimum score to get that class
14000-14060 Light	44000-44999 Dimension arrays and fill with appropriate data
15000-15610 Dive	
16000-16030 Wear	
17000-17020 Un-wear	
18000-18020 Throw	
19000-19130 Push	
20000-20550 Score	
21000-21070 Quit	